# Maintainer: Filipe Laíns (FFY00) <lains@archlinux.org>

_pkgname=jeepney-objects
pkgname=python-$_pkgname
pkgver=0.2.0
pkgrel=1
pkgdesc='Create and publish pure python DBus objects using jeepney'
arch=('any')
url='https://github.com/ocaballeror/jeepney-objects'
license=('GPL')
depends=('python-jeepney')
makedepends=('python-setuptools')
checkdepends=('python-pytest-runner' 'python-pytest-cov' 'dbus')
source=("$pkgname-$pkgver.tar.gz::$url/archive/v$pkgver.tar.gz")
sha512sums=('c639e9e738e43fe26f40deacda87ddd97cac8db870f35597c30f8d96f165ab882441a472cd99f35530107a41bb8bb46618d894fca0e59dd59d037c4a8761dcad')

build() {
  cd $_pkgname-$pkgver

  python setup.py build
}

check() {
  cd $_pkgname-$pkgver

  eval $(dbus-launch --sh-syntax)

  python setup.py pytest
}

package() {
  cd $_pkgname-$pkgver

  python setup.py install --root="$pkgdir" --optimize=1 --skip-build
}

